COMP 410
Fall 2025
Midterm Practice Exam #2 (Solutions)

This is representative of the kinds of topics and kind of questions you may be asked on the midterm. This practice exam, along with assignments 3-4 and the in-class handouts on introductory Prolog, unification in Prolog, repetition and structures in Prolog, lists in Prolog, and the first half of the handout on recursion without accumulators, are intended to be comprehensive of everything on the exam. That is, I will not ask anything that's not somehow covered by those sources.

You are permitted to bring two 8.5 x 11 sheets of paper into the exam with you, and these may have anything on them, either printed or handwritten. Both sides of both sheets can be used.

Unification

Consider each of the following unification attempts. If the unification succeeds, record any values any variables take. If the unification fails, say so.

1.) `foo(1, X) = foo(Y, 2)`

`X = 2, Y = 1`

2.) `foo(1, X) = foo(X, 2)`

`false`

3.) `foo(1, _) = foo(X, 2)`

`X = 1`

4.) `foo(1, _) = foo(1, _)`

`true`

5.) `foo(1, 2, bar) = foo(X, _, _, _)`

```
false
```

6.) `foo(bar(baz), X) = foo(Y, Z), Y = bar(A)`

```
X = Z, Y = bar(baz), A = baz
```

Recursion Without Lists

7.) Consider the following mathematical definition of a recursive function:

$$f_n = \begin{cases} 2 & \text{if } n = 0 \\ 3 & \text{if } n = 1 \\ (3 \times f_{n-1}) + (4 \times f_{n-2}) & \text{otherwise} \end{cases}$$

Write an equivalent definition in Prolog.

```
f(0, 2).
f(1, 3).
f(N, Result) :-
    N > 1,
    MinOne is N - 1,
    MinTwo is N - 2,
    f(MinOne, T1),
    f(MinTwo, T2),
    Result is (3 * T1) + (4 * T2).
```

8.) Write a procedure named `evensBetween`, which will nondeterministically produce all the even numbers within an inclusive range. As a hint, a number `N` is even if and only if the clause `0 is mod(N, 2)` is true. An example query is below:

```
?- evensBetween(1, 4, Even).
Even = 2 ;
Even = 4.
```

```
evensBetween(Min, Max, Min) :-
    Min   =< Max,
    0 is mod(Min, 2).
evensBetween(Min, Max, Result) :-
    Min   < Max,
    NewMin is Min + 1,
    evensBetween(NewMin, Max, Result).
```

9.) Define a procedure named `isPrime` which will determine if a given input number is prime. You may introduce any helpers you wish. Example queries follow:

```
?- isPrime(2).
true .
?- isPrime(3).
true .
?- isPrime(4).
false.
```

As a hint, the following Java-like code:

```
int x = y % z;
```

...is equivalent to the following Prolog code:

```
X is mod(Y, Z)
```

```
isPrime(Num) :-
    FirstTest is Num - 1,
    isPrime(Num, FirstTest).

% isPrime: Number, CurrentTest
isPrime(_, 1).
isPrime(Num, Test) :-
    Test > 1,
    NonZero is mod(Num, Test),
    NonZero \== 0,
    NewTest is Test - 1,
    isPrime(Num, NewTest).
```

Prolog Lists

Consider each of the following unification attempts involving lists. If the unification succeeds, record any values any variables take. If the unification fails, say so.

10.) `[1, 2, _] = [A, B, C|D]`

`A = 1, B = 2, D = []`

11.) `A = [1, 2|B], B = [4]`

`A = [1, 2, 4], B = [4]`

12.) `[[A|B], C] = [[1, 2]|D]`

`A = 1, B = [2], D = [C]`

13.) `X = [A|[2]]`

`X = [A, 2]`

14.) `[A, [B, [C|D]]] = [1, [2, [3, 4]]]`

`A = 1, B = 2, C = 3, D = [4]`

15.) Consider the following inductive list definition, which makes use of Prolog atoms and structures:

$$e \in \mathit{ListElement}$$

$$\ell \in \mathit{List} ::= \mathbf{cons}(e, \ell) \mid \mathbf{nil}$$

Now consider the following unifications, using Prolog lists. Rewrite these unifications using the above definition.

15.a.) `X = [1, 2, 3]`

```
X = cons(1, cons(2, cons(3, nil)))
```

15.b.) `X = [Y|Z]`

```
X = cons(Y, Z)
```

15.c.) `X = [A|[2]]`

```
X = cons(A, cons(2, nil))
```

15.d.) `X = [1, [2, [3]]]`

```
X = cons(1, cons(cons(2, cons(cons(3, nil), nil)), nil))
```

Recursion with Lists

16.) Write a procedure named `allEqual` which will succeed if all list elements are equal to each other according to unification (=). You may introduce any helpers you wish. Example calls are below:

```
?- allEqual([]).
true.
?- allEqual([1, 1, 1]).
true.
?- allEqual([1, 2, 3]).
false.
?- allEqual([1, X, 1]).
X = 1.
?- allEqual([A, B]).
A = B.
?- allEqual([X, 1, 2]).
false.
```

```
allEqual([]).
allEqual([_]).
allEqual([H, H|Rest]) :-
    allEqual([H|Rest]).
```

17.) Write a procedure named `zip`, which takes two lists of the same length, an output list of the same length. The output list is a list of `pair` structures, where each `pair` holds an element from each list, preserving order. If the lists are not the same length, `zip` should fail, though you shouldn't need to explicitly check the length. Example calls are below:

```
?- zip([], [], Output).
Output = [].
?- zip([hello], [goodbye], Output).
Output = [pair(hello, goodbye)].
?- zip([1, 2, 3], [a, b, c], Output).
Output = [pair(1, a), pair(2, b), pair(3, c)].
?- zip([A, B], [C, D], Output).
Output = [pair(A, C), pair(B, D)].
?- zip([foo], [bar, baz], Output).
false.
?- zip([foo, bar], [baz], Output).
false.
```

```
zip([], [], []).
zip([H1|T1], [H2|T2], [pair(H1, H2)|Rest]) :-
    zip(T1, T2, Rest).
```

18.) We can represent a binary tree in Prolog using the following:

- `leaf`: At atom representing a leaf node
- `internal(tree, value, tree)`: A structure recursively containing two trees and some integer value

Write a procedure named sumTree that will compute the sum of all the elements in a given binary tree. Leaf nodes are defined to have a sum of 0. Example queries are shown below:

```
?- sumTree(leaf, Sum).
Sum = 0.
?- sumTree(internal(leaf, 5, leaf), Sum).
Sum = 5.
?- sumTree(internal(internal(leaf, 1, leaf),
                    2,
                    internal(leaf, 3, leaf)),
           Sum).
Sum = 6.
```

```
sumTree(leaf, 0).
sumTree(internal(Left, Value, Right), Sum) :-
  sumTree(Left, LeftSum),
  sumTree(Right, RightSum),
  Sum is LeftSum + Value + RightSum.
```